3 Experimenten

- biologie ——— 1
- wiskunde ——— 19
- natuurkunde — 33
- scheikunde —— 65

Biologie

Simulatie van een populatie

$1 \leq \alpha \leq 4$
is de *groeifactor*

dit is de
logistieke functie

$$p_{n+1} = \alpha p_n (1 - p_n)$$

$0 \leq p \leq 1$
is een relatieve
populatie

p_0 is de initiële
populatie

Biologie

Analytische benadering

een stabiele populatie is een fixpunt van:

$$p = \alpha p(1-p)$$

$$\Longrightarrow \begin{cases} p_f = 0 \\ p_F = (\alpha-1)/\alpha \end{cases}$$

Biologie

Beperkingen op α

voor $\alpha < 1$:

$p_F < 1 \Rightarrow$ enkel p_f geldt

voor $\alpha > 4$:

$\exists p \ni \alpha p(1-p) > 1$

zeker p = 0.5 want
dan is p(1-p) maximaal

Biologie

Populatie in functie van α

```
lim(alfa): (alfa-1)/alfa
:<function lim>
for(alfa:1.0,alfa:=alfa+0.1,alfa<4.01,
    display(alfa,',',lim(alfa),eoln))
:1,0
:1.1,0.0909091
:1.2,0.166667
:1.3,0.230769
:1.4,0.285714

:3.9,0.74359
:4,0.75
```

transcript

Biologie

p_F in functie van α
(met behulp van Excel)

voor een gegeven $p_0 > 0$ en α verwachten we dit als eindpopulatie



Biologie

experimentele controle

```
simul(alfa,p,n):
  for(g:0,g:=g+1,g<n,
      display(g,' -> ',p:=alfa*p*(1-p),eoln))
:<function simul>
simul(1.5,0.5,10)
:0 -> 0.375
:1 -> 0.351562
:2 -> 0.341949
:3 -> 0.33753
:4 -> 0.335405
:5 -> 0.334363
:6 -> 0.333847
:7 -> 0.33359
:8 -> 0.333461
:9 -> 0.333397
```

voor $\alpha = 1.5$
is $p_F = 1/3$

transcript

Biologie

experimentele controle

```
simul(1.0,0.5,100)
:0 -> 0.25
:1 -> 0.1875
:2 -> 0.152344
:3 -> 0.129135
:4 -> 0.112459
:5 -> 0.0998122

:94 -> 0.00986369
:95 -> 0.0097664
:96 -> 0.00967102
:97 -> 0.00957749
:98 -> 0.00948576
:99 -> 0.00939578
```

voor $\alpha = 1.0$
is $p_F = 0$
trage convergentie

transcript

Biologie

relaxatieperiode

```
simul(alfa,p,r,n):
   {for(g:0,g:=g+1,g<r,p:=alfa*p*(1-p));
    for(g:0,g:=g+1,g<n,
        display(g,' -> ',p:=alfa*p*(1-p),eoln))}
:<function simul>
simul(1.0,0.5,5000,10)
:0 -> 0.00019955
:1 -> 0.00019951
:2 -> 0.00019947
:3 -> 0.00019943
:4 -> 0.00019939
:5 -> 0.000199351
:6 -> 0.000199311
:7 -> 0.000199271
:8 -> 0.000199231
:9 -> 0.000199192
```

eerst wordt over r generaties gerelaxeerd

transcript

Biologie

relaxatieperiode (vervolg)

```
simul(1.5,0.1,100,5)
:0 -> 0.333333
:1 -> 0.333333
:2 -> 0.333333
:3 -> 0.333333
:4 -> 0.333333
:
:
simul(1.5,0.9,100,5)
:0 -> 0.333333
:1 -> 0.333333
:2 -> 0.333333
:3 -> 0.333333
:4 -> 0.333333
:
:
```

onafhankelijk van $p_0 > 0$

transcript

Biologie

relaxatieperiode (vervolg)

```
simul(3.2,0.5,100,10)
:0 -> 0.799455
:1 -> 0.513045
:2 -> 0.799455
:3 -> 0.513045
:4 -> 0.799455
:5 -> 0.513045
:6 -> 0.799455
:7 -> 0.513045
:8 -> 0.799455
:9 -> 0.513045
:
:
```

in plaats van te stabiliseren op
$p_F = 0.6875$
alterneert de populatie tussen
twee stabiele waarden

Biologie

relaxatieperiode (vervolg)

```
simul(3.5,0.5,100,10)
:0 -> 0.874997
:1 -> 0.38282
:2 -> 0.826941
:3 -> 0.500884
:4 -> 0.874997
:5 -> 0.38282
:6 -> 0.826941
:7 -> 0.500884
:8 -> 0.874997
:9 -> 0.38282
:
:
```

in plaats van te stabiliseren op
$p_F = 0.71428571429$
alterneert de populatie tussen
vier stabiele waarden

transf

Biologie

relaxatieperiode (vervolg)

```
simul(3.9,0.5,100,30)
:0 -> 0.168915
:1 -> 0.547493
:2 -> 0.966203
:3 -> 0.127353
:4 -> 0.433422
:5 -> 0.957713
:6 -> 0.157946
:7 -> 0.518696

:25 -> 0.184546
:26 -> 0.586906
:27 -> 0.945545
:28 -> 0.200811
:29 -> 0.625895
:
:
```

in plaats van te stabiliseren op
$p_F = 0.74358974359$
is de populatie onvoorspelbaar
= chaos!

Biologie

visualisatie

```
simul(alfa):
   {map[100]: false;
    p: 0.5;
    for(g:0,g:=g+1,g<100,p:=alfa*p*(1-p));
    for(g:0,g:=g+1,g<300,
        {p:= alfa*p*(1-p);
         map[1+trunc(100*p)]:= true});
    display(alfa);
    for(i:1,i:=i+1,i<100,
        if(map[i],display('.',i),void));
    display(eoln)}
:<function simul>
```

transcript

Biologie

visualisatie(vervolg)

```
simul(3.2)
:3.2,52,80
:
simul(3.5)
:3.5,39,51,83,88
:
simul(3.9)
:3.9,10,11,12,13,14,15,16,17,18,19,20,21,22,
23,24,26,27,28,29,30,31,33,34,35,36,37,38,39
,40,41,42,43,44,45,46,47,48,49,50,52,53,54,
55,56,57,58,59,60,62,63,64,65,66,67,68,70,71
,72,74,75,76,77,78,79,80,81,82,83,84,85,86,
87,88,89,90,91,92,93,94,95,96,97,98
```

transcript

Biologie

visualisatie(vervolg)

```
simul():
   {map[100]: false;
    for(alfa:0.0,alfa:=alfa+0.01,not(alfa>4.0),
        {p: 0.5;
         for(g:0,g:=g+1,g<100,p:=alfa*p*(1-p));
         for(g:0,g:=g+1,g<150,
             {p:= alfa*p*(1-p);
              map[1+trunc(100*p)]:=true});
         display(alfa);
         for(i:1,i:=i+1,i<100,
             if(map[i],
                {display(',',i);
                 map[i]:= false},
                void));
         display(eoln)})}
:<function simul>
```

transcript

Biologie



Biologie



Determinant van een matrix

$$\det \begin{bmatrix} 1.4 & -0.7 \\ 2.11.9 & \end{bmatrix} = 4.13$$

Determinant van een matrix (vervolg)

```
matrix: [[1.4,-0.7],[2.1,1.9]]
:[[1.4, -0.7], [2.1, 1.9]]
det2(m):
   {r1: m[1];
    r2: m[2];
    r1[1]*r2[2]-r1[2]*r2[1]}
:<function det2>
det2(matrix)
:4.13
```

transcript

Determinant van een matrix (vervolg)

$$\det \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} = 1$$

$$\det \begin{bmatrix} 1 & 2 & 3 \\ 2 & 1 & 2 \\ 3 & 3 & 1 \end{bmatrix} = 12$$

Determinant van een matrix (vervolg)

```
matrix: [[1,0,0],[0,1,0],[0,0,1]]
:[[1, 0, 0], [0, 1, 0], [0, 0, 1]]
det2(p,q,r,s): p*s-q*r
:<function det2>
det3(m):
   {r1: m[1];
    r2: m[2];
    r3: m[3];
    r1[1]*det2(r2[2],r2[3],r3[2],r3[3])-
    r1[2]*det2(r2[1],r2[3],r3[1],r3[3])+
    r1[3]*det2(r2[1],r2[2],r3[1],r3[2])}
:<function det3>
det3(matrix)
:1
det3([[1,2,3],[2,1,2],[3,3,1]])
:12
```

transcript

Determinant van een matrix (vervolg)

indien $M \in L(\mathbb{R}^n, \mathbb{R}^n)$

$$\Rightarrow$$

$$\det M = \sum_{\pi \in \pi(1,2,\ldots,n)} (-1)^{ord(\pi)} \prod_{i=1}^{n} M_{i\pi_i}$$

voor n = 3 is $\det(M) = M_{11}M_{22}M_{33} - M_{11}M_{23}M_{31} + M_{12}M_{21}M_{33} - M_{13}M_{21}M_{32} + M_{12}M_{23}M_{31} - M_{13}M_{22}M_{31}$

Determinant van een matrix (vervolg)

$$\det\begin{bmatrix}1 & 2 & 3\\ 2 & 1 & 2\\ 3 & 3 & 1\end{bmatrix} = \det\begin{bmatrix}1 & 23\\ 0 & -3 & -4\\ 0 & -3 & -8\end{bmatrix}$$

$$= \det\begin{bmatrix}1 & 23\\ 0 & -3 & -4\\ 0 & 0 & -4\end{bmatrix}$$

$$= 12$$

door triangularisatie
(indien D bovendiagonaalmatrix
geldt det(D) = ∏[diag(D)])

Triangularisatie van een matrix

$$\begin{bmatrix} m_{11} & m_{12} & m_{13} \\ m_{21} & m_{22} & m_{23} \\ m_{31} & m_{32} & m_{33} \end{bmatrix} \quad \begin{array}{|c|c|} -m_{21}/m_{11} & -m_{31}/m_{11} \\ 1 & \\ & 1 \end{array}$$

$$\Downarrow$$

$$\left[\begin{array}{ccc} m_{11} & m_{12} & m_{13} \\ 0 & m_{22}-m_{21}m_{12}/m_{11} & m_{23}-m_{21}m_{13}/m_{11} \\ 0 & m_{32}-m_{31}m_{12}/m_{11} & m_{33}-m_{31}m_{13}/m_{11} \end{array}\right] \quad \begin{array}{|c} -P/Q \\ 1 \end{array}$$

P

Q

$$\Downarrow$$

Triangularisatie van een matrix

$$\begin{bmatrix} 1 & 0 & 0 \\ -2 & 1 & 0 \\ -1 & -1 & 1 \end{bmatrix} \times \begin{bmatrix} 1 & 2 & 3 \\ 2 & 1 & 2 \\ 3 & 3 & 1 \end{bmatrix} = \begin{bmatrix} 1 & 2 & 3 \\ 0 & -3 & -4 \\ 0 & 0 & -4 \end{bmatrix}$$

$\det = 1$

$\det = ?$

$\det = 12$

Triangularisatie van een matrix (vervolg)

```
matrix: [[1,2,3],[2,1,2],[3,3,1]]
:[[1, 2, 3], [2, 1, 2], [3, 3, 1]]
n: size(matrix)
:3
combine(r1,r2):
  {fact: r2[1]/r1[1];
   for(i:n,i:=i-1,not(i=0),
       r2[i]:= r2[i]-r1[i]*fact)}
:<function combine>
combine(matrix[1],matrix[2])
:[0, -3, -4]
combine(matrix[1], matrix[3])
:[0, -3, -8]
matrix
:[[1, 2, 3], [0, -3, -4], [0, -3, -8]]
```

eerste poging

transcript

Triangularisatie van een matrix (vervolg)

```
matrix: [[1,2,3],[2,1,2],[3,3,1]]
:[[1, 2, 3], [2, 1, 2], [3, 3, 1]]
n: size(matrix)
:3
combine(r1,r2,k,n):
  {fact: r2[k]/r1[k];
   for(i:n,i:=i-1,not(i<k),
       r2[i]:= r2[i]-r1[i]*fact)}
:<function combine>
combine(matrix[1],matrix[2],1,n)
:[0, -3, -4]
combine(matrix[1],matrix[3],1,n)
:[0, -3, -8]
combine(matrix[2],matrix[3],2,n)
:[0, 0, -4]
matrix
:[[1, 2, 3], [0, -3, -4], [0, 0, -4]]
```

tweede poging

transcript

Triangularisatie van een matrix (vervolg)

```
matrix: [[1,2,3],[2,1,2],[3,3,1]]
:[[1, 2, 3], [2, 1, 2], [3, 3, 1]]
tri(M):
   {n: size(M);
    for(k:1,k:=k+1,k<n,
        {R1: M[k];
         for(r:k+1,r:=r+1,not(r>n),
             {R2: M[r];
              fact: R2[k]/R1[k];
              for(i:n,i:=i-1,not(i<k),
                  R2[i]:= R2[i]-R1[i]*fact)})});
    M}
:<function tri>
tri(matrix)
:[[1, 2, 3], [0, -3, -4], [0, 0, -4]]
```

consolidatie

transcript

Determinant van een matrix (vervolg)

```
det([[1,2,3],[2,4,2],[3,3,1]])
```

opletten voor deling door nul in:

```
...fact: R2[k]/R1[k]...
```

trix
)

finale oplossing

```
det(M):
   {n: size(M);
    D: 1;
    stop: false;
    for(k:1,k:=k+1,and(not(stop),k<n),
        {R1: M[k];
         stop:= (R1[k]=0);
         for(m:k+1,m:=m+1,and(not(stop),not(m>n)),
             {R1:=M[m];
              stop:=(R1[k]=0);
              if(stop,void,
                 {M[m]:= M[k]; M[k]:= R1; D:= -D})});
         if(stop,
            void,
             {D:= D*R1[k];
              for(r:k+1,r:=r+1,not(r>n),
                  {R2: M[r];
                   fact: R2[k]/R1[k];
                   for(i:n,i:=i-1,not(i<k),
                       R2[i]:= R2[i]-R1[i]*fact})})});
    if(stop,0,{R: M[n]; D*R[n]})}
```

transcript

Determinant van een matrix (besluit)

slotopmerking:

- dit algoritme vraagt een aantal stappen ~ n^3
- het oorspronkelijke algoritme vraagt een aantal stappen ~ n!

trilling van een veer

k > 0

$$m\frac{d^2x}{dt^2}(t) = -kx(t)$$

noteer: x"(t)





trilling van een veer (vervolg)

$$x''(t) = -\omega^2 x(t) \qquad \begin{cases} x(0) = 1 \\ x'(0) = 0 \end{cases}$$

$$\Rightarrow x(t) = \cos\omega t$$

voor de eenvoud

gedwongen trilling van een veer

uitwendige periodische kracht

$$x''(t) = -\omega^2 x(t) + P\cos\Omega t \qquad \Omega \neq \omega$$

$$\Rightarrow x(t) = a\cos(\omega t - \varphi) + x^*(t)$$

algemene oplossing van de homogene vergelijking

bijzondere oplossing van de niet-homogene vergelijking

gedwongen trilling van een veer (vervolg)

$$x''(t) = -\omega^2 x(t) + P\cos\Omega t \qquad \Omega \neq \omega$$

stel $x^*(t) = A\cos\Omega t$

$$\begin{cases} x(0) = 1 \\ x'(0) = 0 \end{cases}$$

$$\Rightarrow -\Omega^2 A\cos\Omega t = -\omega^2\cos\Omega t + P\cos\Omega t$$

$$\Rightarrow A = P/(\omega^2 - \Omega^2)$$

gedwongen trilling van een veer (vervolg)

$$x''(t) = -\omega^2 x(t) + P\cos\Omega t \quad \Omega \neq \omega$$

$$\begin{cases} x(0) = 1 \\ x'(0) = 0 \end{cases}$$

$$\Rightarrow x(t) = (1 - A)\cos\omega t + A\cos\Omega t$$

$$A = \frac{P}{\omega^2 - \Omega^2}$$

```
P: 1
:1
omega: 2
:2
Omega: 3
:3
A: P/(omega^2-Omega^2)
:-0.2
x(t): (1-A)*cos(omega*t)+A*cos(Omega*t)
:<function x>
for(t:0,t:=t+0.1,t<20,display(t,' ',x(t),eoln))
:0 1
:0.1 0.985013
:0.2 0.940206
:0.3 0.866081
:0.4 0.763577
:0.5 0.634215
:0.6 0.48027
```

transcript

gedwongen trilling van een veer (vervolg)



Graf 2.3.6 uit:
ftp://sunsite.doc.ic.ac.uk/

gedwongen trilling van een veer (vervolg)




```
process(omega,Omega,P):
 {A: P/(omega^2-Omega^2);
  x(t): (1-A)*cos(omega*t)+A*cos(Omega*t);
  for(t:0,t:=t+0.1,t<20,display(t,' ',x(t),eoln))}
```

resonantie?

gedwongen trilling van een veer (vervolg)



$A = -2.44...$

$A_{max} = (1+2.44)+2.44 = 5.88$

$A_{min} = 1$

met een groter tijdsbereik...

gedwongen trilling van een veer (vervolg)




```
process(omega,P):
   {Amax(Omega):
      {A: P/(omega^2-Omega^2);
       1+2*abs(A)};
    for(Omega:omega-1,Omega:=Omega+0.03,Omega<omega+1,
        display(Omega,' ',Amax(Omega),eoln))}
```

resonantiepiek

gedrag bij resonantie

$$x''(t) = -\omega^2 x(t) + P\cos\omega t$$

$$\begin{cases} x(0) = 1 & \surd \\ x'(0) = 0 & ? \end{cases}$$

$$\Rightarrow x(t) = \cos\omega t + At\sin\omega t$$

$$A = ?$$

gedrag bij resonantie (vervolg)

$$x'(t) = (A-\omega)\sin\omega t + \omega A t\cos\omega t$$

$$x''(t) = (2A-\omega)\omega\cos\omega t - \omega^2 A t\sin\omega t$$

$$\Rightarrow A = \frac{P}{2\omega} \quad \text{en} \quad x'(0) = 0 \ \surd$$

gedrag bij resonantie (vervolg)




```
process(omega,P):
   {A: P/(2*omega);
    x(t): cos(omega*t)+A*t*sin(omega*t);
    for(t:0,t:=t+0.2,t<100, display(t,' ',x(t),eoln))}
```

trilling van niet-lineaire veren

$k > 0$

$|k'| \ll k$

$$m\frac{d^2x}{dt^2}(t) = -kx(t) + k'x^3(t)$$

de veerkracht is bijna lineair
en symmetrisch in x



trilling van een niet-lineaire veer (vervolg)

$$x''(t) = -\omega^2 x(t) + \varepsilon x^3(t) + P\cos\Omega t$$

$$\begin{cases} x(0) = 1 \\ x'(0) = 0 \end{cases}$$

niet meer analytisch op te lossen!

voor de eenvoud

trilling van een niet-lineaire veer (vervolg)

kwalitatief, benaderd onderzoek:
methode van de eerste harmoniek

beschouw een eerste benadering van de oplossing van de vorm:

$$\xi(t) = A\cos\Omega t$$

trilling van een niet-lineaire veer (vervolg)

$$\xi''(t) = -\Omega^2 A\cos\Omega t$$

$$-\omega^2\xi(t) = -\omega^2 A\cos\Omega t$$

$$\varepsilon\xi^3(t) = \frac{\varepsilon A^3}{4}(3\cos\Omega t + \cos 3\Omega t)$$

verwaarlozen!

trilling van een niet-lineaire veer (vervolg)

$$\Rightarrow \frac{3\varepsilon}{4} A^3 + (\Omega^2 - \omega^2)A + P = 0$$

dus de amplitude van de oscillatie is (benaderend) oplossing van een derdegraadsvergelijking!

wortels van een 3e graadsveelterm

$$x^3 + px + q = 0$$

geval #1: $p=0$

$$\Rightarrow \begin{cases} x_1 = -q^{\frac{1}{3}} & \in \mathbb{R} \\ x_2 = \frac{1}{2} x_1(-1 + \sqrt{3}\,i) & \in \mathbb{C} \\ x_3 = \frac{1}{2} x_1(-1 - \sqrt{3}\,i) & \in \mathbb{C} \end{cases}$$

wortels van een 3e graadsveelterm (vervolg)

```
poly3(p,q):
  if(p=0,
     {x1: -(q^(1/3));
      x2: [-x1/2,-sqrt(3)*x1/2];
      x3: [-x1/2,+sqrt(3)*x1/2];
      display('wortels = ',x1,', ',x2,', ',x3)},
     display('p≠0, dit kan ik nog niet'))
:<function poly3>
poly3(0,1)
:wortels = -1, [0.5, 0.866025], [0.5, -0.866025]
```

transcript

wortels van een 3e graadsveelterm (vervolg)

$$x^3 + px + q = 0$$

stel $x = m\cos\theta$ met $m,\theta \in \mathbb{C}$

$$\Rightarrow$$

$$m^3\cos^3\theta + pm\cos\theta + q = 0$$

$$\updownarrow$$

$$4\cos^3\theta - 3\cos\theta - \cos 3\theta = 0$$

wortels van een 3e graadsveelterm (vervolg)

$$\frac{m^3}{4} = \frac{pm}{-3} = \frac{q}{-\cos 3\theta}$$

$$\Rightarrow$$

$$m = 2\sqrt{\frac{-p}{3}} \qquad \cos 3\theta = \frac{3q}{pm}\,\alpha$$

wortels van een 3e graadsveelterm (vervolg)

geval #2:

$p<0 \Rightarrow m,\alpha \in \mathbb{R}$ en $\alpha \in [-1,+1]$

$$\Rightarrow \theta = \frac{1}{3}\text{bgcos}\alpha$$

$m\cos q,$
$m\cos(q+\frac{2\pi}{3}),$
$m\cos(q+\frac{4\pi}{3})$

$$\Rightarrow \begin{cases} x_1 = m\cos\theta & \in \mathbb{R} \\ x_2 = \frac{1}{2}m(-\cos\theta+\sqrt{3}\sin\theta) & \in \mathbb{R} \\ x_3 = \frac{1}{2}m(-\cos\theta-\sqrt{3}\sin\theta) & \in \mathbb{R} \end{cases}$$

wortels van een 3e graadsveelterm

```
poly3(p,q):
  if(p=0,

     if(p>0,
        display('p>0, dit kan ik nog niet'),
        {m: 2*sqrt(-p/3);
         alfa: 3*q/(p*m);
         if(alfa>1,
            display('alfa>1, dit kan ik nog niet'),
            if(alfa < -1,
               display('alfa<-1, dit kan ik nog niet'),
               {theta: arccos(alfa)/3;
                x1: m*cos(theta);
                x2: m*(-cos(theta)+sqrt(3)*sin(theta))/2;
                x3: m*(-cos(theta)-sqrt(3)*sin(theta))/2;
                display('wortels: ',x1,', ',x2,', ',x3)})))))
:<function poly3>
poly3(-3,1)
:wortels: 1.53209, 0.347296, -1.87939
```

transcript

wortels van een 3e graadsveelterm (vervolg)

geval #3:

$p<0 \Rightarrow m,\alpha \in \mathbb{R}$ en $\alpha > +1$

$\text{bgcosh}\zeta = \ln[\zeta + \sqrt{(\zeta^2-1)}]$

$\sin\xi = -i\sinh(i\xi)$

$\cos\xi = \cosh(i\xi)$

$$\Rightarrow \theta = -\frac{1}{3} i\, \text{bgcosh}\alpha = -i\phi$$

$$\Rightarrow \begin{cases} x_1 = m\cosh\phi & \in \mathbb{R} \\ x_2 = \frac{1}{2}m(-\cosh\phi + \sqrt{3}\, i\, \sinh\phi) & \in \mathbb{C} \\ x_3 = \frac{1}{2}m(-\cosh\phi - \sqrt{3}\, i\, \sinh\phi) & \in \mathbb{C} \end{cases}$$

```
cosh(x): (exp(x)+exp(-x))/2
:<function cosh>
sinh(x): (exp(x)-exp(-x))/2
:<function sinh>
arccosh(x): log(x+sqrt(x^2-1))
:<function arccosh>
```

```
poly3(p,q):
  if(p=0,

     if(p>0,
        display('p>0, dit kan ik nog niet'),
        {m: 2*sqrt(-p/3);
         alfa: 3*q/(p*m);
         if(alfa>1,
            {fi: arccosh(alfa)/3;
             x1: m*cosh(fi);
             x2: tab(-m*cosh(fi)/2, +m*sqrt(3)*sinh(fi)/2);
             x3: tab(-m*cosh(fi)/2, -m*sqrt(3)*sinh(fi)/2);
             display('wortels: ',x1,', ',x2,', ',x3)},
            if(alfa < -1,

               display('wortels: ',x1,', ',x2,', ',x3)))))))
:<function poly3>
poly3(-3,-3)
:wortels: 2.1038, [-1.0519, 0.565236], [-1.0519, -0.565236]
```

transcript

wortels van een 3e graadsveelterm (vervolg)

geval #4:

$p<0 \Rightarrow m,\alpha \in \mathbb{R}$ en $\alpha < -1$

$$\Rightarrow \theta = \frac{1}{3}[\pi - i\,\mathrm{bgcosh}(-\alpha)] = \frac{\pi}{3} - i\phi$$

$$\Rightarrow \begin{cases} x_1 = \frac{1}{2}m(\cosh\phi + \sqrt{3}\,i\,\sinh\phi) & \in \mathbb{C} \\ x_2 = -m\cosh\phi & \in \mathbb{R} \\ x_3 = \frac{1}{2}m(\cosh\phi - \sqrt{3}\,i\,\sinh\phi) & \in \mathbb{C} \end{cases}$$

```
poly3(p,q):
   if(p=0,

      if(p>0,
         display('p>0, dit kan ik nog niet'),
         {m: 2*sqrt(-p/3);
          alfa: 3*q/(p*m);
          if(alfa>1,

             if(alfa < -1,
                {fi: arccosh(-alfa)/3;
                 x1: [m*cosh(fi)/2, +m*sqrt(3)*sinh(fi)/2];
                 x2: -m*cosh(fi);
                 x3: [m*cosh(fi)/2, -m*sqrt(3)*sinh(fi)/2];
                 display('wortels: ',x1,', ',x2,', ',x3)},

                 display('wortels: ',x1,', ',x2,', ',x3)})))))
:<function poly3>
poly3(-3, 3)
:wortels: [1.0519, 0.565236], -2.1038, [1.0519, -0.565236]
```

wortels van een 3e graadsveelterm (vervolg)

geval #5:

$p>0 \Rightarrow m,\alpha \in i\mathbb{R}$

$\text{bgsinh}\zeta = \ln[\zeta + \sqrt{(\zeta^2+1)}]$

$$\Rightarrow \theta = \frac{1}{3}\left[\frac{\pi}{2} + i\,\text{bgsinh}(i\alpha)\right] = \frac{\pi}{6} + i\phi$$

$$\Rightarrow \begin{cases} x_1 = \frac{1}{2}m(\sqrt{3}\cosh\phi - i\sinh\phi) & \in \mathbb{C} \\ x_2 = \frac{1}{2}m(\sqrt{3}\cosh\phi + i\sinh\phi) & \in \mathbb{C} \\ x_3 = -i\,m\sinh\phi & \in \mathbb{R} \end{cases}$$

```
arcsinh(x): log(x+sqrt(x^2+1))
:<function arcsinh>
```

```
poly3(p,q):
  if(p=0,
     {x1: -(q^(1/3));
      x2: [-x1/2,-sqrt(3)*x1/2];
      x3: [-x1/2,+sqrt(3)*x1/2];
      display('wortels = ',x1,', ',x2,', ',x3)},
     if(p>0,
        {im: -2*sqrt(p/3);
         ialfa: -3*q/(p*im);
         fi: arcsinh(ialfa)/3;
         x1: [-im*sinh(fi)/2, +im*sqrt(3)*cosh(fi)/2];
         x2: [-im*sinh(fi)/2, -im*sqrt(3)*cosh(fi)/2];
         x3: im*sinh(fi);
         display('wortels: ',x1,', ',x2,', ',x3)},

                  display('wortels: ',x1,', ',x2,', ',x3)}))})
:<function poly3>
poly3(3,3)
:wortels: [0.408866, -1.87123], [0.408866, 1.87123], -0.817732
```

```
poly3(p,q):
  if(p=0,
     -(q^(1/3)),
     if(p>0,
        {im: -2*sqrt(p/3);
         ialfa: -3*q/(p*im);
         fi: arcsinh(ialfa)/3;
         im*sinh(fi)},
        {m: 2*sqrt(-p/3);
         alfa: 3*q/(p*m);
         if(alfa>1,
            {fi: arccosh(alfa)/3;
             m*cosh(fi)},
            if(alfa<-1,
               {fi: arccosh(-alfa)/3;
                -m*cosh(fi)},
               {theta: arccos(alfa)/3;
                [m*cos(theta),
                 m*(-cos(theta)+sqrt(3)*sin(theta))/2,
                 m*(-cos(theta)-sqrt(3)*sin(theta))/2]})))})
```

enkel reële wortels

transcript

ling van een niet-lineaire veer (vervolg)

```
amplitude(epsilon,omega,P):
  {a: 3*epsilon/4;
   for(Omega:omega-1,Omega:=Omega+0.01,Omega<omega+1,
       {b: (Omega^2-omega^2);
        A: poly3(b/a, P/a);
        if(is_table(A),
           display(Omega,' ',A[1],' ',A[2],' ',A[3],eoln),
           display(Omega,' ',A,' ',0,' ',0,eoln))})}
```

$\omega = 2,\ \varepsilon = 0.01,\ P = 0.5$



de periodieke tabel



	IA												IIIA	IVA	VA	VIA	VIIA	0
1	1 H	IIA																2 He
2	3 Li	4 Be											5 B	6 C	7 N	8 O	9 F	10 Ne
3	11 Na	12 Mg	IIIB	IVB	VB	VIB	VIIB		VIII		IB	IIB	13 Al	14 Si	15 P	16 S	17 Cl	18 Ar
4	19 K	20 Ca	21 Sc	22 Ti	23 V	24 Cr	25 Mn	26 Fe	27 Co	28 Ni	29 Cu	30 Zn	31 Ga	32 Ge	33 As	34 Se	35 Br	36 Kr
5	37 Rb	38 Sr	39 Y	40 Zr	41 Nb	42 Mo	43 Tc	44 Ru	45 Rh	46 Pd	47 Ag	48 Cd	49 In	50 Sn	51 Sb	52 Te	53 I	54 Xe
6	55 Cs	56 Ba	57 * La	72 Hf	73 Ta	74 W	75 Re	76 Os	77 Ir	78 Pt	79 Au	80 Hg	81 Tl	82 Pb	83 Bi	84 Po	85 At	86 Rn
7	87 Fr	88 Ra	89 + Ac	104 Rf	105 Ha	106 106	107 107	108 108	109 109	110 110								

* Lanthanide Series	58 Ce	59 Pr	60 Nd	61 Pm	62 Sm	63 Eu	64 Gd	65 Tb	66 Dy	67 Ho	68 Er	69 Tm	70 Yb	71 Lu
+ Actinide Series	90 Th	91 Pa	92 U	93 Np	94 Pu	95 Am	96 Cm	97 Bk	98 Cf	99 Es	100 Fm	101 Md	102 No	103 Lr

de periodieke tabel databank

```
init():
   {db[105]: 0;
    set(at_nr,sym,naam,at_gew,kolom,rij):
      db[at_nr]:= [at_nr,sym,naam,at_gew,kolom,rij];
    set(  1,'H' ,'Waterstof'     ,  1,'Ia'  ,1);
    set(  2,'He','Helium'        ,  4,'0'   ,1);
    set(  3,'Li','Lithium'       ,  7,'Ia'  ,2);
    set(  4,'Be','Berylium'      ,  9,'IIa' ,2);
    set(  5,'B' ,'Borium'        , 11,'IIIa',2);
    set(  6,'C' ,'Koolstof'      , 12,'IVa' ,2);
    set(  7,'N' ,'Stikstof'      , 14,'Va'  ,2);
    set(  8,'O' ,'Zuurstof'      , 16,'VIa' ,2);
    set(  9,'F' ,'Fluor'         , 19,'VIIa',2);
    set( 10,'Ne','Neon'          , 20,'0'   ,2);
    set( 11,'Na','Natrium'       , 23,'Ia'  ,3);
    set( 12,'Mg','Magnesium'     , 24,'IIa' ,3);
    set( 13,'Al','Aluminium'     , 27,'IIIa',3);

    set(102,'No','Nobelium'      ,254,'ACT' ,0);
    set(103,'Lr','Lawrencium'    ,257,'ACT' ,0);
    set(104,'Rf','Rutherfordium', -1,'IVb' ,7);
    set(105,'Ha','Hahnium'       , -1,'Vb'  ,7);
    db}
:<function init>
```

atoomnummer

symbool

naam

rij in de tabel

kolom in de tabel

atoomgewicht

de periodieke tabel databank

```
DB: init()
:[[1, H, Waterstof, 1, Ia, 1], [2, He, Helium, 4, 0, 1], [3, Li, Lithium, 7, Ia, 2], [4, Be, Ber
9, IIa, 2], [5, B, Borium, 11, IIIa, 2], [6, C, Koolstof, 12, IVa, 2], [7, N, Stikstof, 14, Va,
O, Zuurstof, 16, VIa, 2], [9, F, Fluor, 19, VIIa, 2], [10, Ne, Neon, 20, 0, 2], [11, Na, Natrium
Ia, 3], [12, Mg, Magnesium, 24, IIa, 3], [13, Al, Aluminium, 27, IIIa, 3], [14, Si, Silicium, 28, IVa,
3], [15, P, Fosfor, 31, Va, 3], [16, S, Zwavel, 32, VIa, 3], [17, Cl, Chloor, 35, VIIa, 3], [18, Ar,
Argon, 40, 0, 3], [19, K, Kalium, 39, Ia, 4], [20, Ca, Calcium, 40, IIa, 4], [21, Sc, Scandium, 45,
IIIb, 4], [22, Ti, Titanium, 48, IVb, 4], [23, V, Vanadium, 51, Vb, 4], [24, Cr, Chroom, 52, VIb, 4],
[25, Mn, Mangaan, 55, VIIb, 4], [26, Fe, Ijzer, 56, VIII, 4], [27, Co, Kobalt, 59, VIII, 4], [28, Ni,
Nikkel, 58, VIII, 4], [29, Cu, Koper, 63, Ib, 4], [30, Zn, Zink, 64, IIb, 4], [31, Ga, Gallium, 69,
IIIa, 4], [32, Ge, Germanium, 74, IVa, 4], [33, As, Arseen, 75, Va, 4], [34, Se, Selenium, 80, VIa, 4],
[35, Br, Broom, 79, VIIa, 4], [36, Kr, Krypton, 84, 0, 4], [37, Rb, Rubidium, 85, Ia, 5], [38, Sr,
Strontium, 88, IIa, 5], [39, Y, Yttrium, 89, IIIb, 5], [40, Zi, Zirkonium, 90, IVb, 5], [41, Nb,
Niobium, 93, Vb, 5], [42, Mo, Molybdeen, 98, VIb, 5], [43, Tc, Technetium, 98, VIIb, 5], [44, Ru,
Ruthenium, 102, VIII, 5], [45, Rh, Rhodiumm, 103, VIII, 5], [46, Pd, Palladium, 106, VIII, 5], [47, Ag,
Zilver, 107, Ib, 5], [48, Cd, Cadmium, 114, IIb, 5], [49, In, Indium, 115, IIIa, 5], [50, Sn, Tin, 120,
IVa, 5], [51, Sb, Antimoon, 121, Va, 5], [52, Te, Tellurium, 130, VIa, 5], [53, I, Jodium, 127, VIIa,
5], [54, Xe, Xenon, 132, 0, 5], [55, Cs, Caesium, 133, Ia, 6], [56, Ba, Barium, 138, IIa, 6], [57, La,
Lanthanium, 139, IIIb, 6], [58, Ce, Cerium, 140, LAN, 0], [59, Pr, Praseadimium, 141, LAN, 0], [60, Nd,
Neodimium, 142, LAN, 0], [61, Pm, Promethium, 147, LAN, 0], [62, Sm, Samarium, 152, LAN, 0], [63, Eu,
Europium, 153, LAN, 0], [64, Gd, Gadolinium, 158, LAN, 0], [65, Tb, Terbium, 159, LAN, 0], [66, Dy,
Dysprosium, 164, LAN, 0], [67, Ha, H
Thalium, 169, LAN, 0], [70, Yb, Ytte
Hafnium, 180, IVb, 6], [73, Ta, Tant
Rhenium, 187, VIIb, 6], [76, Os, Osmi
Platina, 195, VIII, 6], [79, Au, Goud, 197, Ib, 6], [80, Hg, Kwik, 202, IIb, 6], [81, Tl, Thallium,
205, IIIa, 6], [82, Pb, Lood, 208, IVa, 6], [83, Bi, Bismuth, 209, Va, 6], [84, Po, Polonium, 210, VIa,
6], [85, At, Astatium, 210, VIIa, 6], [86, Rn, Radon, 222, 0, 6], [87, Fr, Francium, 223, Ia, 7], [88,
Ra, Radium, 226, IIa, 7], [89, Ac, Actinium, 227, IIIb, 7], [90, Th, Thorium, 232, ACT, 0], [91, Pa,
Protactinium, 231, ACT, 0], [92, U, Uranium, 238, ACT, 0], [93, Np, Neptunium, 237, ACT, 0], [94, Pu,
Plutonium, 239, ACT, 0], [95, Am, Americium, 243, ACT, 0], [96, Cm, Curium, 245, ACT, 0], [97, Bk,
Berkelium, 247, ACT, 0], [98, Cf, Californium, 251, ACT, 0], [99, Es, Einsteinium, 254, ACT, 0], [100,
Fm, Fermium, 252, ACT, 0], [101, Md, Mendelevium, 256, ACT, 0], [102, No, Nobelium, 254, ACT, 0], [103,
Lr, Lawrencium, 257, ACT, 0], [104, Rf, Rutherfordium, -1, IVb, 7], [105, Ha, Hahnium, -1, Vb, 7]]
```

DB: init()

transcript

patroonherkenning in de databank



patroonherkenning in de databank

```
match(?,?,?,?,'0',?)
:[2, He, Helium, 4, 0, 1]
:[10, Ne, Neon, 20, 0, 2]
:[18, Ar, Argon, 40, 0, 3]
:[36, Kr, Krypton, 84, 0, 4]
:[54, Xe, Xenon, 132, 0, 5]
:[86, Rn, Radon, 222, 0, 6]
:
match(?,?,?,?,'VIII',4)
:[26, Fe, Ijzer, 56, VIII, 4]
:[27, Co, Kobalt, 59, VIII, 4]
:[28, Ni, Nikkel, 58, VIII, 4]
:
match(?,?,?,124,?,?)
:
match(?,?,?,180,?,?)
:[72, Hf, Hafnium, 180, IVb, 6]
:[73, Ta, Tantalium, 180, Vb, 6]
```

transcript

patroonherkenning in de databank (vervolg)

```
?: void
:<void>
match(at_nr,sym,naam,at_gew,kolom,rij):
  for(at:1,at:=at+1,at<106,
      {atoom: DB[at];
       if(and(or(is_void(at_nr),at_nr=atoom[1]),
              and(or(is_void(sym),sym=atoom[2]),
                  and(or(is_void(naam),naam=atoom[3]),
                      and(or(is_void(at_gew),at_gew=atoom[4]),
                          and(or(is_void(kolom),kolom=atoom[5]),
                              or(is_void(rij),rij=atoom[6])))))),
          display(atoom, eoln),
          eoln)})
:<function match>
```

transcript

patroonherkenning in de databank (vervolg)

```
match@patroon:
   {check(atoom,k):
      if(or(is_void(patroon[k]),patroon[k]=atoom[k]),
         if(k<6,
            check(atoom,k+1),
            true),
         false);
    for(at:1,at:=at+1,at<106,
        if(check(DB[at],1),
           display(DB[at],eoln),
           eoln))}
:<function match>
```

beschouw de patroon als tabel

check doorloopt elk atoom

transcript

patroonherkenning in de databank

```
test(x): and(x>40,x<120)
:<function test>
match(?,?,?,test,'IIIa',?)
:[31, Ga, Gallium, 69, IIIa, 4]
:[49, In, Indium, 115, IIIa, 5]
:
test1(t): and(not(t<'N'),t<'O')
:<function test1>
test2(x): (x\\2)=0
:<function test2>
match(test2,test1,?,?,?,?)
:[10, Ne, Neon, 20, 0, 2]
:[28, Ni, Nikkel, 58, VIII, 4]
:[60, Nd, Neodimium, 142, LAN, 0]
:[102, No, Nobelium, 254, ACT, 0]
```

alle atomen uit IIIa met atoomgewicht tussen 40 en 120

alle atomen met een even atoomnummer en een symbool dat begint met N

transcript

patroonherkenning in de databank (vervolg)

```
match@patroon:
   {check(atoom,k):
      if(k=7,
         true,
         {P: patroon[k];
          if(is_void(P),
             check(atoom,k+1),
             if(is_function(P),
                if(P(atoom[k]),
                   check(atoom, k+1),
                   false),
                if(P=atoom[k],
                   check(atoom, k+1),
                   false)))});
    for(at:1, at:=at+1, at<106,
        if(check(DB[at],1),
           display(DB[at], eoln),
           eoln))}
:<function match>
```

indien P_k een functie is, bereken dan $P_k(A_k)$

transcript